Supplementary material to:

# Relaxation optimized double acquisition (RODA) as an alternative for virtual decoupling of NMR spectra

Alvar D. Gossert*, Gerhard Wider

Department of Biology, Biomolecular NMR spectroscopy platform, ETH Zürich, 8093 Zürich, Switzerland

*Corresponding author: alvar.gossert@biol.ethz.ch




**Figure S1. Pulse scheme used for ($^{15}$N,$^{1}$H) isotope filtering and RODA spin state selection.** On the lines labelled $^{1}$H and $^{15}$N, representing the respective radio-frequency (rf) channels, narrow and wide bars stand for non-selective 90° and 180° rf-pulses, respectively. Phases are x, unless indicated otherwise above the pulse bar. The $^{1}$H selective pulses (curved shapes) are used to maintain the water magnetization along the +z axis. The relevant *J* coupling is $^{1}J_{NH}$ = 93 Hz. For inversion of $^{15}$N spins a $90_x$-$240_y$-$90_x$ composite pulse was chosen because of its wide inversion profile. To suppress the strong detergent signals in the region between 1 to 5 ppm in the measurements with the sample containing mixed OmpX/DHPC micelles, the refocusing 180° pulse on $^{1}$H was replaced by a REBURP pulse of a duration of 1 ms centred at 8.5 ppm to selectively invert the amide proton spins. Isotope filtering is accomplished using $\phi_1$ = y, -y and $\phi_2$ = x, x, -x, -x and spin-state selection is achieved with $\phi_3$ = 45°, 135°, 225°, 315°; $\phi_4$ = x, -x and $\phi_{rec1}$ = x, x, -x, -x and RODA $\phi_{rec2}$ = x, -x. Pulsed field gradients (PFGs): $G_1$ = 40 G/cm, 1 ms; $G_2$ = 9 G/cm, 0.8 ms; $G_3$ = 35 G/cm, 0.3 ms; $G_4$ = 40 G/cm, 0.3 ms; $G_5$ = 21 G/cm, 0.3 ms. Water suppression was achieved by a WATERGATE sequence [1].




**Figure S2. Pulse schemes used for 2D [$^{19}$F,$^{13}$C] and [$^{1}$H,$^{13}$C] ST2-PT (top) and RODA (bottom) experiments** [2–4]. Same schematic representation as in supplementary figure 1. The relevant *J*-coupling is $^{1}J_{CH}$ (≈ 160 Hz) or $^{1}J_{FC}$ (≈ 240 Hz). For ST2-PT the phase cycle is the following: $\phi_1$ = y, -y, x, -x and $\phi_1$' = -y, y, x, -x (for quadrature detection, see below); $\phi_2^D$ = -y ; $\phi_2^U$ = y; $\phi_3^D$ = y ; $\phi_3^U$ = -y, where the superscript D and U denote selection of the downfield or the upfield component, respectively, where $\phi_2$ acts on the $\delta_1$ and $\phi_3$ on $\delta_2$ dimension. $\phi_4$ = x and $\phi_{rec}$ = x, -x, -y, y. Quadrature detection in the $F_1$ dimension is performed using the echo-anti echo method. For each $t_1$ increment $\phi_1$ and $\phi_{rec}$ are inverted, and two scans are recorded with alternating $\phi_1$ and $\phi_1$' phases and simultaneous inversion of $\phi_2$ and $\phi_4$. Pulsed field gradients (PFGs) along the z-axis: $G_1$ = 5.5 G/cm, 0.6 ms; $G_2$ = 15.5 G/cm.

For the corresponding RODA pulse sequence (bottom) the phase cycle is: $\phi_1$ = 4 x, 4 -x; $\phi_2^D$ = 315°, 135°; $\phi_2^U$ = 45°, 225°; $\phi_3^D$ = -y, -y, y y; $\phi_3^U$ = y, y, -y, -y; $\phi_4$ = x, -x, -x, x and $\phi_{rec1}$ = $\phi_{rec2}$ = x, -x, x, -x, -x, x, -x, x. Quadrature detection in the $F_1$ dimension is performed using the States-TPPI method. For each $t_1$ increment two scans are recorded with $\phi_1$ incremented by 90°. PFGs along the z-axis: $G_1$ = 30.0 G/cm, 0.6 ms; $G_2$ = 22.5 G/cm; $G_2$ = 18.5 G/cm. The grey pulses at the end of the sequence can be inserted for obtaining a similar baseline in both scans if distortions are observed.

## Pulse program 1:
## 2D [$^{1}$H,$^{1}$H$^{N}$]-NOESY with $^{15}$N editing in 2$^{nd}$ dimension, and TROSY component selection with ($^{15}$N)-$^{1}$H RODA

(Pulse program for spectrum shown in Fig. 3 of main text)

```
; 2D [1H,1HN]-NOESY with 15N editing in F2, and TROSY component selection with (15N)-1H RODA

; A.D. Gossert, G. Wider, J. Magn. Res. (2022)

;optimization of water flip back:
;- optimize water flipdown (sp1,ph11,ph18)
;- optimize water flipup (sp2,ph26)
;Attention: Pulse sequence is not prosol compatible


;pl1   : power for 1H
;pl2   : power for 13C
;pl3   : power for 15N

;sp1   : water flipup power
;sp2   : water flipdown power
;spnam1: gauss128_5
;spnam2: gauss128_5

;p1    : 90 degree hard pulse 1H
;p3    : 90 degree hard pulse 13C
;p4    : 180 degree hard pulse 13C (225d for 5/600)
;p5    : 90 degree hard pulse 15N
;p11   : water flipback pulse (1.5m)

;p21   : 900u (Gradient in first INEPT)
;p22   : 900u (Gradient in second INEPT)
;p23   : 400u (Gradient in 1st WG)
;p24   : 900u (z-filter Gradient)
;p25   : 400u (Gradient in 2nd WG)
;gpz1  : 80%
;gpz2  : 35%
;gpz3  : 70%
;gpz4  : 80%
;gpz5  : 70%

;d1    : relaxation delay (1 s)
;d2    : 1/2J delay (5.4 ms)
;d4    : 1/2J delay (5.4 ms)
;d5    : delay 3-9-19=1/(Hz between nulls) (170–250 us)

;in0   : 1/(2 SW) (Hz)

define delay DEL11
define delay DEL12
define delay DEL21
define delay DEL22
define delay DEL23
define delay DEL2

#include <Avance.incl>

#define GRADIENT1   10u p21:gp1 200u
#define GRADIENT2   10u p22:gp2 200u
#define GRADIENT3   10u p23:gp3 200u
#define GRADIENT4   10u p24:gp4 200u
```

```
#define GRADIENT5   10u p25:gp5 200u

#define WG3919    p1*0.2308 ph20 d5 p1*0.6923 ph20 d5 p1*1.4615 ph20 d5 p1*1.4615 ph22 d5 p1*0.6923 ph22
d5 p1*0.2308 ph22

"p2=2*p1"
"p4=2*p3"
"p8=2*p5"
"p6=8*(p5/3)"

"DEL11=(d2/2)-(p22+210u)-p5-p11-10u"
"DEL12=(d2/2)-(p22+210u)-p5-p11-10u"
"DEL21=(d2/4)-(p11+10u)-p6/2-p5+80u"
"DEL22=(d2/4)-(p11+10u)-p6/2-p5+80u"
"DEL23=6u+p5+p5+p6"

"d0=in0/2-(p1*2/3.1415)"

"l2 = 1"



1  10u ze
2  1m
5  10u
   d1
   20u pl1:f1
   20u pl2:f2
   20u pl3:f3
   20u LOCKH_ON
;---------------------------------------- 1H evolution
   (refalign (p1 ph8 d0 d0 p1 ph25):f1 center (p8 ph20):f3 center (p3 ph20 2u p4 ph21 2u p3 ph20):f2)
;---------------------------------------- NOE mixing time
   d7
   (p5 ph20):f3
   GRADIENT1
   10u
;---------------------------------------- 15N-editing element
   (p1 ph1):f1
   10u
   (p11:sp1 ph11:r):f1 (p5 ph20 3u p6 ph21 3u p5 ph24):f3
   GRADIENT2
   DEL11
   (center (p2 ph20):f1 (p5 ph20 3u p6 ph21 3u p5 ph20):f3)
   GRADIENT2
   DEL12
   (refalign (p11:sp1 ph12:r):f1 ralign (p5 ph20 3u p6 ph21 3u p5 ph24):f3)
   10u pl1:f1
   (p1 ph2):f1
;---------------------------------------- z-filter gradient against remaining H
   GRADIENT4
   (p1 ph3):f1
;---------------------------------------- 1/4J separate I13 I24
   (lalign (p5 ph20 3u p6 ph21 3u p5 ph26):f3 (10u p11:sp1 ph13:r):f1)
   DEL21 pl1:f1
   (center (p2 ph20):f1 (p5 ph20 3u p6 ph21 3u p5 ph20):f3)
   DEL22 pl1:f1
   (ralign (p5 ph20 3u p6 ph21 3u p5 ph26):f3 (p11:sp1 ph14:r 10u pl1):f1)
   (p1 ph4:r):f1
;---------------------------------------- Watergate
   GRADIENT3
   (WG3919):f1
   GRADIENT3
;---------------------------------------- 1st acquisition I13
```

```
   goscnp ph31
;---------------------------------------- Clean up
   GRADIENT5
;---------------------------------------- Record second component
   (p11:sp2 ph15:r):f1
   10u pl1:f1
   (p1 ph20):f1
   (p5 ph20 3u p6 ph21 3u p5 ph20):f3
   GRADIENT5
   (WG3919):f1
   GRADIENT5
   DEL23
;---------------------------------------- 2nd acquisition I13
   go=2 ph30
   1m mc #0 to 2
       F1PH(ip8, id0)

10u do:f1
10u do:f2
10u do:f3
10u do:f4
10u LOCKH_OFF
exit

;----------------------------------- Phases


ph1 =    1 1 3 3
ph2 =    0 2 0 2 2 0 2 0
ph3 =(8) 1 3 5 7
ph4 =    1
ph8 =    0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2

ph11=    3 3 1 1
ph12=    2 0 2 0 0 2 0 2
ph13=(8) 5 7 1 3
ph14=    3
ph15=    2


ph20=    0
ph21=    1
ph22=    2
ph23=    3

ph24=    0 0 0 0 2 2 2 2
ph25=(8) 1
ph26=    2 0


ph30=3 1 3 1 3 1 3 1 1 3 1 3 1 3 1 3
ph31=0 0 0 0 2 2 2 2 2 2 2 2 0 0 0 0
```

## Pulse program 2:
## 1D ($^{19}$F)-$^{13}$C RODA

(Pulse program for spectrum shown in Fig. 4 of main text)

```
;  1D (19F)-13C RODA

; A.D. Gossert, G. Wider, J. Magn. Res. (2022)

;- NBL : 2
;- ZGOPTNS: -DDOWNFIELD: select downfield component, default selects upfield component


;Channels for (19F)-13C-RODA
;pl1   : power for 13C
;pl2   : power for 19F
;pl3   : power for 1H

;p1    : 90 degree hard pulse 13C
;p3    : 90 degree hard pulse 19F

;p16   : 1000u (long gradient)
;p19   : 300u (short gradient)
;gpz1  : 60%
;gpz2  : 35%
;gpz3  : 47%

;d1    : relaxation delay (1s)
;d2    : 1/2J delay

;cnst2 : J-coupling in Hz (240 Hz for 1J(13C,19F))

define delay DEL11
define delay DEL12
define delay DEL21
define delay DEL22
define delay DEL23

#include <Avance.incl>

#define GRADIENT1   10u p16:gp1 500u
#define GRADIENT2   10u p19:gp2 150u
#define GRADIENT3   10u p16:gp3 200u

"p2=2*p1"
"p4=2*p3"

"d2=1/2*cnst2"

"DEL21=(d2/4)-(p19+160u)-(larger(p2,p4))/2"
"DEL22=(d2/4)-(p19+160u)-(larger(p2,p4))/2"
"DEL23=p4"



1  10u ze
2  30m LOCKH_OFF
   10u do:f3
   10u do:f1
   d1
   10u st0
   20u pl1:f1
   20u pl2:f2
   20u pl13:f3
   20u LOCKH_ON
;---------------------------------------- Purge z magnetisation on F
5  10u
   (p3 ph20):f2
   GRADIENT1
;---------------------------------------- 1/4J to separate I13 I24
   (p1 ph1):f1
```

```
   GRADIENT2
   DEL21
   (center (p2 ph20):f1 (p4 ph20):f2)
   GRADIENT2
   DEL22
   (p1 ph2:r):f1
;---------------------------------------- 1st acquisition I13
   goscnp ph30 cpd3:f3
;---------------------------------------- Clean up
   10u do:f3
   GRADIENT3
   10u st
   20u pl13:f3
;---------------------------------------- Read I24 and convert to I13
   (p1 ph3):f1
   (p4 ph20):f2
   (p2 ph4):f1
   DEL23
   (p1 ph21):f1
;---------------------------------------- 2nd acquisition I13
   go=2 ph31 cpd3:f3
   10u do:f3
   10m wr #0


10u LOCKH_OFF
10u do:f1
10u do:f2
10u do:f3
10u LOCKH_OFF
exit

;---------------------------------------- Phases

#ifdef DOWNFIELD
   ph1 =(8) 7 3          ; Selection of downfield component
   ph2 =    3 3 1 1
#else
   ph1 =(8) 1 5          ; Selection of upfield component
   ph2 =    1 1 3 3
#endif

ph3 = 0 2 2 0
ph4 = 1 0

ph20= 0
ph21= 1
ph22= 2
ph23= 3

ph30= 0 2

ph31= 0 2
```

## Pulse program 3:
## 2D [$^{19}$F,$^{13}$C]-HSQC-RODA and 2D [$^{1}$H,$^{13}$C$^{aro}$]-HSQC-RODA

(Pulse program for spectrum shown in Fig. 5B of main text)

```
; 2D [19F,13C]-HSQC-RODA and 2D [1H,13Caro]-HSQC-RODA

; A.D. Gossert, G. Wider, J. Magn. Res. (2022)

;- Switch channels 2 and 3 for 2D [1H,13Caro]-HSQC-RODA
;- ZGOPTNS: -DDOWNFIELD: select downfield component, default selects upfield component

;pl1   : power for 13C
;pl2   : power for 19F
;pl3   : power for 1H

;p1    : 90 degree hard pulse 13C
;p3    : 90 degree hard pulse 19F

;p16   : 1000u (long gradient pulse)
;p19   : 300u (short gradient pulse)
;gpz1  : 60%
;gpz2  : 35%
;gpz3  : 47%

;d1    : relaxation delay (1s)
;d2    : 1/2J delay

;cnst2 : J-coupling in Hz

define delay DEL11
define delay DEL12
define delay DEL21
define delay DEL22
define delay DEL23
define delay DELTA

#include <Avance.incl>

#define GRADIENT0   10u p16:gp0 500u
#define GRADIENT1   10u p19:gp1 200u
#define GRADIENT2   10u p19:gp2 150u
#define GRADIENT3   10u p16:gp3 200u

"p2=2*p1"
"p4=2*p3"

"d2=1/2*cnst2"

"d0=3u"
"in0=inf1/2"
"DELTA=d0*2+p2"

"DEL11=(d2/2)-(p19+210u)-(larger(p2,p4))/2"
"DEL12=(d2/2)+2*d0+p2-(p19+210u)-(larger(p2,p4))/2"
"DEL21=(d2/4)-(p19+160u)-(larger(p2,p4))/2" ;consider adding baseopt echo delay
"DEL22=(d2/4)-(p19+160u)-(larger(p2,p4))/2"
"DEL23=p4"


1  10u ze
2  30m LOCKH_OFF
   10u do:f3
   10u do:f1
   d1
   20u pl1:f1
   20u pl2:f2
   20u pl13:f3
   20u LOCKH_ON
;---------------------------------------- Purge z magnetisation on 13C
5  10u
```

```
   (p1 ph20):f1
   GRADIENT0
;---------------------------------------- First 1/2J INEPT from 19F to 13C
   (p3 ph10):f2
   d0
   (p2 ph20):f1
   d0
   GRADIENT1
   DEL11
   (center (p4 ph20):f2 (p2 ph20):f1 )
   GRADIENT1
   DEL12
   (p3 ph21):f2
;---------------------------------------- Refocussing 1/4J INEPT to separate I13 I24
#ifdef DOWNFIELD
   (p1 ph11):f1
#else
   (p1 ph1):f1
#endif
   GRADIENT2
   DEL21
   (center (p2 ph20):f1 (p4 ph20):f2)
   GRADIENT2
   DEL22
#ifdef DOWNFIELD
   (p1 ph12:r):f1
#else
   (p1 ph2:r):f1
#endif
;---------------------------------------- 1st acquisition I13
   goscnp ph30 cpd3:f3
;---------------------------------------- Clean up
   10u do:f3
   GRADIENT3
   20u pl13:f3
;---------------------------------------- Read I24 and convert to I13
   (p1 ph3):f1
   (p4 ph20):f2
   (p2 ph4):f1
   DEL23
   (p1 ph21):f1
;---------------------------------------- 2nd acquisition I13
   go=2 ph31 cpd3:f3
   10u do:f3
   30m mc #0 to 2
                        F1PH(calph(ph10, +90), caldel(d0, +in0))


10u LOCKH_OFF
10u do:f1
10u do:f2
10u do:f3
10u LOCKH_OFF
exit

;---------------------------------------- Phases
ph10=    0 0 0 0 2 2 2 2
ph1 =(8) 1 5                    ; Selection of upfield component
ph11=(8) 7 3                    ; Selection of downfield component
ph2 =    1 1 3 3
ph12=    3 3 1 1                ; Inverted phase for correct phase in downfield component case

ph3 =    0 2 2 0
ph4 =    0 1

ph20=0
ph21=1
ph22=2
ph23=3

ph30=0 2 0 2 2 0 2 0

ph31=0 2 0 2 2 0 2 0
```

## Pulse program 4:
## 2D [$^{19}$F,$^{13}$C]-TROSY, 2D [$^{1}$H,$^{13}$C]-TROSY and 2D [$^{1}$H,$^{15}$N]-TROSY

(Pulse program for spectrum shown in Fig. 5 of main text)

```
; 2D [I,X]-TROSY with component selection by means of ST2PT, X-detected
; I = 19F or 1H, X = 13C or 15N.

; A.D. Gossert, G. Wider, J. Magn. Res. (2022)
; K. Takeuchi, H. Arthanari, I. Shimada, G. Wagner, J Biomol NMR. 63 (2015) 323–331


;-ZGOPTNS:
;               Component selection: -DF1U -DF2D
;               F1 = Dimension 1, Up- or Downfield component
;
;                       FC-TROSY: F1U, F2U
;                       HN-TROSY: F1D, F2U
;                       HC-TROSY: F1D, F2D
;                       default FC-TROSY F1U, F2U
;
;               Component selection depends on several factors, test on your spectrometer, if correct
component is selected


;for FC TROSY: f1:13C, f2:19F, f3:1H, f4:2H
;pl1   : power for 13C
;pl2   : power for 19F
;pl3   : power for 1H


;p1    : 90 degree hard pulse f1 (13C)
;p3    : 90 degree hard pulse f2 (19F)

;p16   : 1000u (long gradient pulse)
;p19   : 300u (short gradient pulse)
;gpz0: 3%
;gpz1: 30%
;gpz2: 80%
;gpz3: 45%
;gpz4: 50%
;gpz5: 16.2%
;gpnam0: SINE.100
;gpnam1: SINE.100
;gpnam2: SINE.50
;gpnam3: SINE.100
;gpnam4: SINE.100
;gpnam5: SINE.50

;d0 : incremented delay (2D)                        [6 usec]
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                            [30 msec]
;d16: delay for homospoil/gradient recovery

;d26 : 1/(4J)YH

;cnst4: = J(YH)
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;FnMODE: echo-antiecho


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>


"p2=p1*2"
"p4=p3*2"
```

```
"d11=30m"
"d12=20u"

"d21=1s/(cnst4*4)"

"d0=3u"

"DELTA1=d21-p16-d16-4u"
"DELTA2=d21-p1-p16-d16-4u"



"in0=inf1/2"


"l0=1"


1 d11 ze
  d11 LOCKDEC_ON
  50u LOCKH_ON
  d11 H2_PULSE
  d11 pl17:f4

2 d11 do:f4 do:f3 do:f2
3 d11 H2_LOCK
  9m LOCKH_OFF
  d1 pl1:f1
  50u UNBLKGRAD
  d12 H2_PULSE

4 d11 pl1:f1 pl2:f2 pl3:f3

;--------------------- Start on 19F, evolution without decoupling
  if "l0 %2 == 1"
     {
     (p3 ph4):f2
     }
  else
     {
     (p3 ph5):f2
     }

  d0
  d0

;--------------------- First INEPT to 13C
  (p1 ph6)
  4u
  p16:gp3
  d16
  DELTA2 pl1:f1
  (center (p2 ph1) (p4 ph1):f2 )
  4u
  DELTA1  pl16:f3
  p16:gp3
  d16
;--------------------- Refocussing INEPT on 13C
  (center (p1 ph1) (p3 ph8):f2 )
  4u
  p16:gp4
  d16
  DELTA1
  (center (p2 ph1) (p4 ph1):f2 )
  DELTA1 pl16:f3
  p16:gp4
  d16
  4u
  (p3 ph7):f2
  4u BLKGRAMP
;--------------------- Acquistion on 13C
  go=2 ph31 cpd3:f3 cpd4:f4
  d11 do:f2 do:f3 do:f4 mc #0 to 2
```

```
     F1EA(calph(ph6, +180) & calph(ph7, +180) & calclc(l0, 1), caldel(d0, +in0) & calph(ph4, +180) &
calph(ph5, +180) & calph(ph31, +180))

  d11 do:f2 do:f3 do:f4

  d11 H2_LOCK
  d11 LOCKH_OFF
  d11 LOCKDEC_OFF

exit

;---------------------------------------- Phases
ph1=0
ph2=1
ph3=3
ph4=1 3 0 2
ph5=3 1 0 2

#ifdef F1D
        ph6=3
#else
        ph6=1
#endif

#ifdef F2D
        ph8=1
#else
        ph8=3
#endif


ph7=0

ph31=0 2 3 1
```

## Pulse program 5:
## 1D ($^{13}$C')-$^{13}$C$^{ali}$ RODA

(Pulse program for spectrum shown in Fig. 6 of main text)

```
; 1D CaCO RODA

; A.D. Gossert, G. Wider, J. Magn. Res. (2022)

; W. Bermel, I. Bertini,L. Duma, I.C. Felli, L. Emsley, R. Pierattelli,
;   P.R. Vasos, Angew. Chem. Int. Ed. 44, 3089-3092 (2005)



;pl1 : f1 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling

;sp23: f1 channel - shaped pulse  90 degree  (on resonance)
;sp24: f1 channel - shaped pulse 180 degree  (on resonance)
;sp25: f1 channel - shaped pulse  90 degree  (on resonance)
;                   for time reversed pulse
;sp27: f1 channel - shaped pulse 180 degree  (Ca off resonance)

;p11: f1 channel -  90 degree shaped pulse
;p12: f1 channel - 180 degree shaped pulse
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(8J(COCa))                                      [4.5 msec]
;cnst2: CACO J-coupling (in Hz)
;cnst21: CO chemical shift (offset, in ppm)
;cnst22: Calpha chemical shift (offset, in ppm)
;o1p: CO chemical shift (cnst21)

;l0: flag to switch between inphase and antiphase
;inf1: 1/SW(Ca) = 2 * DW(Ca)
;in0: 1/(2 * SW(Ca)) =  DW(Ca)
;nd0: 2

;p16: homospoil/gradient pulse                          [1 msec]
;d16: delay for homospoil/gradient recovery
;gpz1: 50%
;gpnam1: SMSQ10.100


prosol relations=<triple_c>


#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"d11=30m"
"d12=20u"

"d2=1/(8*cnst2)"


"d0=3u"
"d20=d2-p12-4u-p12/2"
"in0=inf1/2"


"spoff23=0"
"spoff24=0"
"spoff25=0"
"spoff26=bf1*((cnst21-cnst22)/1000000)"
"spoff27=bf1*((cnst22-cnst21)/1000000)"
"spoff28=0"
```

```
1 ze
  d11 pl12:f2
2 d11 do:f2
3 d1
  d12 cpd2:f2
  50u UNBLKGRAD
  p16:gp1
  d16 fq=cnst21(bf ppm):f1
;---------------------------------------- 1/4J INEPT to separate I13 I24
  (p11:sp23 ph1)
   4u
  (p12:sp27 ph20)   ;Bloch-Siegert compensation
  d20
  (p12:sp24 ph20)
  4u
  (p12:sp27 ph20)
  d20
  (p11:sp25 ph2)
;---------------------------------------- 1st acquisition I13
   goscnp ph30
;---------------------------------------- Clean up
   p16:gp2
   d16
   10u st
   10u BLKGRAD
;---------------------------------------- Read I24 and convert to I13
   (p12:sp27 ph20):f1
   (p11:sp23 ph3):f1
;---------------------------------------- 2nd acquisition I13
   go=2 ph31
   10m do:f2
   10m wr #0

10u do:f1
10u do:f2
exit



;---------------------------------------- Phases

#ifdef DOWNFIELD
  ph1 =(8) 7 3          ; Selection of downfield component
;  ph2 =    3 1         ; phase for -z
  ph2 =    1 3          ; phase for +z
#else
  ph1 =(8) 1 5          ; Selection of upfield component
;  ph2 =    1 3         ; phase for -z
  ph2 =    3 1          ; phase for +z
#endif


ph3 =0
ph4 =1

ph20=0
ph21=1
ph22=2
ph23=3

ph30=2 0

ph31=0
```

## References


[1] M. Liu, X. Mao, C. Ye, H. Huang, J.K. Nicholson, J.C. Lindon, Improved WATERGATE pulse sequences for solvent suppression in NMR spectroscopy, Journal of Magnetic Resonance. 132 (1998) 125–129. https://doi.org/10.1006/jmre.1998.1405.

[2] K.V. Pervushin, G. Wider, K. Wüthrich, Single Transition-to-single Transition Polarization Transfer (ST2-PT) in [$^{15}$N,$^{1}$H]-TROSY, Journal of Biomolecular NMR. 12 (1998) 345–348. https://doi.org/10.1023/A:1008268930690.

[3] K. Takeuchi, H. Arthanari, I. Shimada, G. Wagner, Nitrogen detected TROSY at high field yields high resolution and sensitivity for protein NMR, J Biomol NMR. 63 (2015) 323–331. https://doi.org/10.1007/s10858-015-9991-y.

[4] A. Boeszoermenyi, S. Chhabra, A. Dubey, D.L. Radeva, N.T. Burdzhiev, C.D. Chanev, O.I. Petrov, V.M. Gelev, M. Zhang, C. Anklin, H. Kovacs, G. Wagner, I. Kuprov, K. Takeuchi, H. Arthanari, Aromatic $^{19}$F-$^{13}$C TROSY: a background-free approach to probe biomolecular structure, function, and dynamics, Nat Methods. 16 (2019) 333–340. https://doi.org/10.1038/s41592-019-0334-x.